STAT 705 Chapters 20 & 21: Randomized complete block designs

Timothy Hanson

Department of Statistics, University of South Carolina

Stat 705: Data Analysis II

21.1 Randomized complete block designs

Subjects placed into homogeneous groups, called *blocks*. All treatment combinations assigned randomly to subjects within blocks.

Example (p. 895): executives exposed to one of three methods (treatment, $i = 1$ utility method, $i = 2$ worry method, $i = 3$ comparison method) of quantifying maximum risk premium they would be willing to pay to avoid uncertainty in a business decision. Response is "degree of confidence" in the method on a scale from 0 (no confidence) to 20 (complete confidence). It is thought that confidence is related to age, so the subjects are blocked according to age ($j = 1, 2, 3, 4, 5$ from oldest to youngest). $n_T = 15$ subjects are recruited, with three subjects in each of the 5 age categories. Within each age category, the three subjects are randomly given one of the three treatments.

RCB designs, comments

- With thoughtful blocking, can provide more precise results than completely randomized design.
- There is only one replication for each pairing of treatment and block; need to assume no interaction between treatments and blocks to obtain estimate of σ^2.
- The blocking variable is observational, not experimental. Cannot infer causal relationship. Not a problem though...usually only care about treatments.

Model and inference

One observation per block/treatment combination gives $n_T = ab$.
Need to fit model IV to get SSE > 0

$$Y_{ij} = \mu + \alpha_i + \beta_j + \epsilon_{ij}.$$

Estimates obtained via LS as usual,

$$Q(\boldsymbol{\alpha}, \boldsymbol{\beta}) = \sum_{i=1}^{a} \sum_{j=1}^{b} (Y_{ij} - [\mu + \alpha_i + \beta_j])^2$$

minimized subject to $\alpha_a = \beta_b = 0$.

ANOVA table

Source	SS	df	MS	F	p-value
A	$SSA = b\sum_{i=1}^{a}(\bar{Y}_{i\bullet} - \bar{Y}_{\bullet\bullet})^2$	$a-1$	$\frac{SSA}{a-1}$	$\frac{MSA}{MSE}$	p_1
B	$SSB = a\sum_{j=1}^{b}(\bar{Y}_{\bullet j} - \bar{Y}_{\bullet\bullet})^2$	$b-1$	$\frac{SSB}{b-1}$	$\frac{MSB}{MSE}$	p_2
Error	$SSE = \sum_{i=1}^{a}\sum_{j=1}^{b}(Y_{ij} - \bar{Y}_{i\bullet} - \bar{Y}_{\bullet j} + \bar{Y}_{\bullet\bullet})^2$	$(a-1)(b-1)$	$\frac{SSE}{(a-1)(b-1)}$		
Total	$SSE = \sum_{i=1}^{a}\sum_{j=1}^{b}(Y_{ij} - \bar{Y}_{\bullet\bullet})^2$	$ab-1$			

Here, $p_1 = P\{F(a-1,(a-1)(b-1)) > \frac{MSA}{MSE}\}$ tests $H_0 : \alpha_1 = \cdots = \alpha_a = 0$ (no blocking effect) and $p_2 = P\{F(b-1,(a-1)(b-1)) > \frac{MSB}{MSE}\}$ tests $H_0 : \beta_1 = \cdots = \beta_b = 0$ (no treatment effect). These appear in SAS as Type III tests.

If reject $H_0 : \beta_j = 0$, then obtain inferences in treatment effects as usual, e.g. `lsmeans B / pdiff adjust=tukey cl;`

Diagnostics

1. Profile (spaghetti) plots of the Y_{ij} vs. treatment j, connected by block i are useful. Should be somewhat parallel if additive model is okay, but there is a lot of sampling variability here as $\hat{\mu}_{ij} = Y_{ij}$.
2. Standard SAS diagnostic panel: e_{ij} vs. $\hat{Y}_{ij}$, normal probability plot of the $\{e_{ij}\}$, etc. Can also look at e_{ij} vs. either i or j, should show constant variance within blocks and treatments.
3. Tukey's test for additivity.

Tukey's test for additivity

Reduced model is additive $Y_{ij} = \mu + \alpha_i + \beta_j + \epsilon_{ij}$. Full model is

$$Y_{ij} = \mu + \alpha_i + \beta_j + D\alpha_i\beta_j + \epsilon_{ij}.$$

This is more restrictive than using a general interaction $(\alpha\beta)_{ij}$, leaves df to estimate error.

$$\hat{D} = \frac{\sum_{i=1}^{a}\sum_{j=1}^{b}(\bar{Y}_{i\bullet} - \bar{Y}_{\bullet\bullet})(\bar{Y}_{\bullet j} - \bar{Y}_{\bullet\bullet})}{\sum_{i=1}^{a}(\bar{Y}_{i\bullet} - \bar{Y}_{\bullet\bullet})^2\sum_{j=1}^{b}(\bar{Y}_{\bullet j} - \bar{Y}_{\bullet\bullet})^2}.$$

$$SSAB^* = \sum_{i=1}^{a}\sum_{j=1}^{b}\hat{D}^2(\bar{Y}_{i\bullet} - \bar{Y}_{\bullet\bullet})^2(\bar{Y}_{\bullet j} - \bar{Y}_{\bullet\bullet})^2,$$

and SSTO=SSA+SSB+SSAB*+SSE*.

$$F* = \frac{SSAB^*}{SSE^*/(ab - a - b)} \sim F(1, ab - a - b),$$

if $H_0 : D = 0$ is true.

Trick for additivity test

let $\hat{Y}_{ij}$ be fitted values from additive model. Fit ANCOVA model

$$Y_{ij} = \mu + \alpha_i + \beta_j + \gamma \hat{Y}_{ij}^2 + \epsilon_{ij}.$$

Test of $H_0 : \gamma = 0$ is same as test of $H_0 : D = 0$, the F-statistics are the same and the p-values are the same.

Confidence ratings, pp. 895-896

```
data conf;
 input rating age$ method$ @@;
datalines;
   1  1  1   5  1  2   8  1  3
   2  2  1   8  2  2  14  2  3
   7  3  1   9  3  2  16  3  3
   6  4  1  13  4  2  18  4  3
  12  5  1  14  5  2  17  5  3
;

proc format;
value $ac '1'='youngest' '2'='age grp II' '3'='age grp III' 4='age grp IV' 5='oldest';
value $mc '1'='utility' '2'='worry' 3='compare';

* first obtain interaction plot by fitting model V;
* trajectories look reasonably parallel;
proc glm data=conf plots=all;
 class age method;
 model rating=age|method;
run;
```

Confidence ratings

```
* fit additive model;
proc glm data=conf plots=all;
 class age method;
 format age $ac. method $mc.;
 model rating=age method / solution;
 output out=tukeytest p=p; * p=yhat values for Tukey's test;
 lsmeans method / pdiff adjust=tukey alpha=0.05 cl;
run;

* Tukey test for additivity;
* p-value=0.79 so model IV is okay;
proc glm data=tukeytest;
 title 'Test for additivity is Type III p*p p-value';
 class age method;
 model rating=age method p*p;
run;
```